
R-Type II 

© 1989 Activision 
 

--------------------------- 

Deprotecting  

MFM + Copylock Protection! 

--------------------------- 

 
Requirements for this Tutorial: 

 
 . Amiga 500 (1 MB Chip) or WinUAE 
 . Assembler (ASM-Pro or similar) 
 . R-Type II Original or 
   CAPS Image (#2065)  
 . Source or Binary of 
   Rob Northens Sectorloader 
   (included in this archive) 

Once again we begin our little journey with trying to 

duplicate the Original Disk: 

 

 
 

We don’t need to wait until the copy is finished…  we 

already notice that this won’t work! ☺ 

 

Let’s see what is happening in the Bootblock, enter 

AR and read the first and only dos-readable Track to 

e.g . address $50000, using “rt 0 1 $50000”. 

 

Let’s disassemble a bit … <d $50000> and scroll down 

some lines… 

 

 
 

Standard Stuff, killing DMA and Interrupts, copying 

the code from $5004e up to address $78000 

and finally executing it with that “RTS” instruction -> 

(Address $78000 has been pushed onto the stack 

before so the RTS will lead us there!) 

 

The code at $5004e already has some interesting 

information for us… we can see the addressregister 

A0 being set with $500, then this value is pushed 

onto the stack,  a routine is called (BSR $50076), and 

finally the “RTS” will execute the code at $500. So I 

believe that the routine at $50076 will surely load 

some data to $500. ☺ 

 

Another interesting thing is the value that is move’d 

into some memory region right before the RTS is  

executed. 

 

 A hardcoded value like #$AE3B9CE3 looks 

suspicious to me so let’s also remember this. 

 

Reset your Amiga now, insert Gamedisk and let 

the loader track about 2 seconds…. Enter AR to 

see what is loaded up to address $500. 

 

 

Ouch.. this looks like a little Copylock-Routine 

which is executed at first. 

But perhaps we don’t need to investigate this 

further, let’s see where the copylock ends. If 

we remember right, the Bootblock just move’d 

that suspicious “Copylock-Like” result to 

address $E2C… a place that is not too far away 

from the copylock start. Let’s disassemble that 

area: 

 

 
Ok, this is surely the code that is executed after 

the Copylockroutine is passed. It compares our  

$AE3B9CE3 with the value returned in D0.  

 

Let us set a breakpoint here to see what D0 

looks like, behave like shown in the next 

picture: 

So after some time we hear the well-known 

trackgrind and our AR will pop up again, telling us 

that the instruction at $E2A is the next one to be 

executed. Let’s have a look at the value in D0 … it is 

what we expected it to be: $AE3B9CE3.  

 

For testing purposes you could restart your Amiga, 

reboot the game and after some tracking, set a 

breakpoint to the start of the Copylock at $500.  

After AR pops up, insert some other disk, set a 

breakpoint at the end of the Copylock again and see 

what is happening to the D0 register. The value will 

be different and the game will show a green screen 

and stop loading! 

 

So getting rid of this Copylock will be quite easy. We 

only need to overwrite the start of the Copylock, 

putting our special value directly to register  D0, 

followed by a jump to the end of the Copylock. 

Due to the fact that we did not decode the inner 

loop of the Copylock, we cannot be sure if this 

special value has perhaps also been moved to some 

other memorylocations . This is happening quite 

often.. so the games can check these addresses later 

on and if not set with the correct value, lead to some 

strange behavior. 

 

We can easily check this by searching for this value 

throughout  chipmemory. Try it like this: 

 

 
 

As we can see, our special value has been 

moved to 4 different places in chipmem.  

($7801E is part of the bootcode which was 

copied into highmem). 

 

We need to remember these addresses as well 

and fill them with the correct value lateron. 

 

Okayyy, this should be the copylock part… now 

let us find out how the Trackloader works to 

obtain a dump from the whole disk. 

 

Leave Action Replay now and let the game 

continue loading…. while the game is tracking 

let’s enter AR again to disassemble the 

trackloading code….. in my case the Program 

Counter was somewhere around $FA00… 

 

 
 

Ok, this looks for sure like some trackloader 

actions here, we are at the right place.  

Lets scroll downwards in memory to obtain the 

beginning of the Loader. 

 
 

The code at address $F960 could be the start of 

the trackloader, right in front are only 

senseless instructions.  

Also the *8 multiplication of D0 (done by 

shiftoperation ASL.W #3, D0) which is then 

used as an offset to address $FB20 looks like some 

filetable calculation to me. ☺ 

 

Let’s see what address $FB20 is all about: 

 

 

Even for none expert eyes this should look like a 

Filetable.  

 

After some more analyzing of the values  

I would clearly say that a file-entry holds the 

following informations: 

 

Offset: 

0: Start-Track .B 

1: Start-Sector on Track .B 

2: Filesize in sectors .W 

4: Filesize in bytes .L 

 

You can see the first parameter in that table goes 

from $2 (File 1) up to $76 (last File), surely a 

Tracknumber. The second bytevalue is always in a 

range between $0 and $B and thus surely a 

sectoroffset. A normal dosdisk consists of 11 sectors,  

each one is 512 ($200) bytes. So an offset of $B (11) 

would make no sense on a dos-formatted disk... In 

this case the gamedisk seems to hold a further sector 

and thus can use $1800 bytes each track instead of 

$1600 on a standard disk. 

 

The upcoming wordvalue in the filetable increases 

with the filesize… we could try an easy calculation to 

prove that this is the sectorsize of a file, e.g. with the 

values of  File #1 and #2: 

 

$243CC / $200 (size of a sector) = $121 ! 

$8BC5   / $200 (size of a sector) =   $45 ! 

 
Seems to be correct. ☺ 

 

Okay… we should be able to call the Loader at 

$F960 now for ourselves, passing some values that 

will load the entire disk for us. 

Lets shortly think about the values we will need: 

 

We will obviously set the first parameter 

(starttrack) to $2 and the sectoroffset to $0. This 

should be the correct start of the disk. 

 

To find out the size of bytes to read we need to 

look at the last file in the table. 

It starts at track $76, sector offset $2 and has a size 

of $4FEF bytes. 

 

The endposition of the disk would then be: 

$76 * $1800 + ($2 * $200) + $4FEF which is 

$B63EF. 

 

So if we wanted to rip the disk from the very first 

track then $B63EF would be our size!!  

 

But due to the fact that the first file starts on track 

$2 and not on $0 we need to subtract 2*$1800 

bytes to get the correct size, which will then be: 

$B33EF !!  

A size of decimal 734191 bytes. 

 

If we remember right, the loader also needs the 

size in sectors, so we need to calc this as well… here 

we go: $B33EF / $200 = $599. 

 

As a standard dosdisk can hold 901120 bytes on 

160 tracks we can easily fit this back onto one disk 

as well. Good to know! 

 

Well it seems as we know everything now to get a 

full diskdump. Lets reset our machine once again, 

wait some seconds and before the trackgrind 

happens, enter AR and set a breakpoint to our 

loader at $F960, catching it when it’s called for the 

first time. ☺ 

 

We will set the filenumber in D0 to #0 and edit the 

first filetableentry so that it fits our needs. Only 

thing we need to do then is to set another 

breakpoint on the end of the trackloader. Instead 

of searching for this place we will grab the return 

address right from the stack and place a new 

breakpoint there. Adressregister A0 holds the 

loading adress of the trackloader, we overwrite this 

value to be $40000. 

Lets get started! Reset, let the game track some 

seconds, then set a breakpoint to the loader at 

$F960. 

 

 
 

When breakpoint happened, edit the values and 

registers like in the picture above and leave AR 

again. The game will (hopefully) start tracking 

the entire disk now, in this case to address 

$40000. 

 

After a minute, loading will be finished and AR 

pops up again. 

 

 
 

The complete disk should be available in memory 

now, ready for being saved!! Insert an empty disk 

and save the dump with  

<sm dump, 40000 40000+$b33ef>. 

 

That’s it! We can care about our crackdisk now and 

can put the original back in it’s case. 

__________________________________________ 

 

 

 

 

 

 

After some testing with the crackimage I noticed 

that the Filetable is missing in the dumped data…  

the game loads it seperately before the trackloader 

is called for the first time. But that’s not a big deal. 

 

If you still remember, the first file in the filetable 

begins on track #2 sector 1. The filetable itself is 

stored in sector #0 right in front. So we already 

grabbed it because we dumped from that position. 

 

We only need to remember this when coding the 

crackimage… we have to copy it into place before 

tracking the image back onto disk ! 

 

So boot up your favourite assembler now, reserve 

about 800 KB Chipmem and begin to code the 

diskimage. The sourcecode is included in this 

archive -> so no need to retype it. 

 

This time I used the well known sectorloader from 

Rob Northen. The binary is included in this tutorial-

archive. 

 

It works with the following parameters: 

 

D0 = Drivenumber (0=DF0:) 

D1 = Startsector 

D2 = Sectors to read 

D3 = Some Drivemotor on / off parameter  

A0 = Loadaddress 

A1 = MFM-Buffer 

 

Enough said, lets type some code! 

 

 
 
The fully commented sourcecode comes up on the 

last 2 pages. 

 

When finished typing or whatever, follow the steps 

that are described at the top of the source to 

create the cracked diskimage. 

Finally reboot your machine and have a play.  

This game rocks!! ☺ 

; #################################################################### 

; ## R-TYPE II (C) ACTIVISON                                        ## 

; ## DISKIMAGE (133 TRACKS) FOR FLASHTRO TUTORIAL.                  ## 

; ## !! USE CHIPMEM FOR THIS FOR WRITEBACK !!                       ## 

; ## -------------------------------------------------------------- ## 

; ## USAGE:                                                         ## 

; ## 1. <A>  ASSEMBLE SOURCECODE                                    ## 

; ## 2. <J>  EXECUTE SOURCE TO MAKE NECCESSARY PATCHES IN DISKIMAGE ## 

; ## 3. <WT> WRITE BACK 133 TRACKS FROM LABEL #DISK_START           ## 

; ## 4. <CC> CALCULATE BOOTBLOCK-CHECKSUM                           ## 

; ## -------------------------------------------------------------- ## 

; ###################################################(ALPHA ONE)###### 

 

 

DISKOFFSET_COPYLOCK_CODE    = 2*$1800+$200   

DISKOFFSET_ORIGINAL_LOADER  = 2*$1800+$200+$F960-$500  

DISKOFFSET_FILETABLE     = 2*$1800+$200+$FB20-$500 

 

 INCDIR "SOURCES:TUTORIAL/" 

 

 ; OVERWRITE OLD COPYLOCK CODE WITH OUR NEW LOADER 

 ; ----------------------------------------------- 

 

 LEA NEW_COPYLOCK_CODE(PC),A0 

 LEA DISK_START+DISKOFFSET_COPYLOCK_CODE,A1 

 MOVE.L #(COPYLOCK_CODE_END-NEW_COPYLOCK_CODE)-1,D0 

COPY_C: MOVE.B (A0)+,(A1)+ 

DBF D0,COPY_C 

 

 ; OVERWRITE FIRST INSTRUCTION OF ORIGINAL LOADER WITH 

 ; A "JMP $500" -> SO OUR LOADER IS EXECUTED! 

 ; --------------------------------------------------- 

 

 LEA DISK_START+DISKOFFSET_ORIGINAL_LOADER,A0 

 MOVE.W #$4EF9,(A0)+   

 MOVE.L #$500,(A0)+ 

 
 ; THE FILETABLE IS LOCATED ON TRACK #2 SECTOR #0 ON DISK 

 ; AND IS LOADED SEPERATELY BY THE ORIGINAL CODE. WE SKIP THIS PART 

 ; AND COPY IT INTO THE RIGHT PLACE HERE ($FB20 IN MEMORY). 

 ; ---------------------------------------------------------------- 

 

 LEA DISK_START+2*$1800,A0   ; TR #2 SEC #0 

 LEA DISK_START+DISKOFFSET_FILETABLE,A1 

 MOVE.L #$90,D0     ; SIZE OF FILETABLE 

COPY_F: MOVE.B (A0)+,(A1)+ 

 DBF D0,COPY_F 

 MOVEQ #0,D0     ; END OF PATCHING!! 

 RTS 

 

; ------------------------------------------------------------------------- 

 

; =========================================================================== 

; THE NEW TRACKLOADER 

; ------------------- 

; HERE COMES OUR NEW LOADER THAT IS WRITTEN TO THE PLACE OF THE 

; ORGINAL COPYLOCK, IN MEMORY AT $500, ON DISK LOCATED ON TRACK #2, SECTOR #1 

; (FILE #1 IN FILETABLE) 

; =========================================================================== 

 

NEW_COPYLOCK_CODE:   ; <- $500 IN MEMORY ! 

 MOVEQ #0,D1   ; HANDLE THE FILES LIKE THE ORIGINAL 

 MOVEQ #0,D2 

 ASL.W #3,D0   ; FILENUMBER * 8 

 LEA $FB20,A1  ; FILETABLE PTR 

 MOVE.B (A1,D0.W),D1  ; D1 = TRACKNUM 

 MULS #$1800,D1  ; * TRACKSIZE $1800 

 MOVE.B 1(A1,D0.W),D2  ; D2 = SECTORNUM 

 MULS.W #512,D2   ; * SECTORSIZE $200 

 ADD.L D2,D1   ; GET CORRECT BYTEOFFSET 

 DIVS.W #512,D1   ; AND TURN IT INTO AN SECTOROFFSET 

 MOVE.W 2(A1,D0.W),D2  ; GET FILESIZE IN SECTORS 

 ADDQ.W #1,D2   ; +1   

 MOVEQ #0,D3     

 MOVE.L 4(A1,D0.W),-(A7) ; SAVE REAL BYTESIZE OF FILE ON STACK  

 LEA $2FD80,A1  ; MFM POINTER 

AGAIN: MOVEQ #0,D0   ; DRIVENUMBER (DF0:) 

 BSR.W TLOADER   ; CALL RNC TRACKLOADER 

 TST.B D0   ; LOADING OK? 

 BEQ.B LD_OK   ; YEAH -> END TLOADER 

 

BLINKER: 

     BTST #6,$BFE001  ; 

 BNE.B FLASH   ; OTHERWISE MAKE SCREEN FLASH 

 BRA.B AGAIN   ; AND WAIT FOR LEFT MOUSE TO  

FLASH: MOVE.W D0,$DFF180  ; REPEAT PROCEDURE :P 

 SUBQ.W #1,D0   ; 

 BRA.B BLINKER   ; 

LD_OK: MOVE.L (A7)+,D0  ; RESTORE BYTESIZE OF FILE 

 RTS    ; FOR FURTHER GAME-USAGE 

  

TLOADER: 

INCBIN "RNCLOADER.BIN" ; ROB NORTHEN SECTORLOADER 

 

COPYLOCK_CODE_END: 

 

; ==<DISKIMAGE>======================================================== 

 

DISK_START: 

 

 DC.B "DOS",0    ; TRACK 0 (BOOT) 

 DC.L 0 

 DC.L 0 

 

 MOVE.W #$2,$1C(A1)   ; READTRACK 

 MOVE.L #2*$1800+$200,$2C(A1)  ; DISKPOSITION OF FILE #1 IN FILETABLE  

 MOVE.L #$24400,$24(A1)  ; SIZE 

 MOVE.L #$50000,$28(A1)  ; LOADADDRESS 

 JSR -$1C8(A6)   ; DOIO() 

 

 LEA COPY_C(PC),A0   ; COPY NEXT PART OF THE  

 LEA $78000,A1   ; BOOTCODE TO $78000 

 MOVE.L #(BOOTEND-COPY_C)-1,D0  ; SO THAT WE DONT OVERWRITE 

COPY: MOVE.B (A0)+,(A1)+   ; OUR OWN BOOTCODE WHEN 

 DBF D0,COPY    ; COPYING GAMEDATA DOWN TO $500 !! 

 

 MOVE.W #$7FFF,D0   ; KILL DMA/INTERRUPTS 

 MOVE.W D0,$DFF096 

 MOVE.W D0,$DFF09A 

 MOVE.L #$78000,$80.W   ; CONTINUE AT $78000 IN  

 TRAP #0    ; SUPERVISOR MODE 

 

COPY_C: LEA $50000,A0   ; COPY OUR GAMEDATA DOWN 

 LEA $500.W,A1   ; FROM $50000 TO $500 !! 

 MOVE.L #$24400,D0 

COPY_D: MOVE.B (A0)+,(A1)+ 

 MOVE.W D0,$DFF180 

 SUBQ.L #1,D0 

 BNE.B COPY_D 

 

 MOVE.L #$4E714E71,$E86.W  ; NOP OUT LOADING OF FILETABLE 

 MOVE.W #$4E71,$E8A.W   ; 

 

 MOVE.L #$AE3B9CE3,D0   ; MOVE COPYLOCK KEY TO D0 

 MOVE.L D0,$60.W   ; AND TO ALL MEMORYLOCATIONS 

 MOVE.L D0,$E2C.W 

 MOVE.L D0,$7FF0C 

 MOVE.L D0,$7FF24 

 MOVE.L D0,$7FFBA 

 JMP $E2A.W    ; NO NEED FOR A JMP TO $500. 

      ; WE GO DIRECTLY TO $E2A! 

 

BOOTEND: 

BLK.B $1800-(BOOTEND-DISK_START),0 ; ZEROES REST OF LONGTRACK 0 

 

 BLK.B $1800,0    ; TRACK 1 (EMPTY) 

 

DUMP: INCBIN "DUMP"    ; TRACK 2 - END 

 

 BLK.B 2577,0    ; ZEROES TO GET AN EVEN DISKSIZE 

      ; 133 DOS TRACKS IN ALL 

 

DISK_END: 

 


